# Maintainer: Maxime Gauduin <alucryd@archlinux.org>

pkgname=libretro-yabause
pkgver=3342
pkgrel=1
pkgdesc='Sega Saturn core'
arch=(loong64 x86_64)
url=https://github.com/libretro/yabause
license=(GPL2)
groups=(libretro)
depends=(
  glew
  glibc
  libretro-core-info
)
makedepends=(git)
_commit=4c96b96f7fbe07223627c469ff33376b2a634748
source=(libretro-yabause::git+https://github.com/libretro/yabause.git#commit=${_commit})
sha256sums=(SKIP)

pkgver() {
  cd libretro-yabause

  git rev-list --count HEAD
}

build() {
  make -C libretro-yabause/yabause/src/libretro
}

package() {
  install -Dm 644 libretro-yabause/yabause/src/libretro/yabause_libretro.so -t "${pkgdir}"/usr/lib/libretro/
}

# vim: ts=2 sw=2 et:
